mizar と集合論

師玉康成

目 次

第1章　TARSKIの公理系

このテキストは，集合論を題材に，mizarを使った，数学定理の記述が，実際にどのようになされているかを説明します。

mizarでは，その数学記述言語によって書かれたテキストをarticleと称します。mizarシステムやそれで使われる数学記述言語の文法については，中村八東教授著のmizar講義録第３版

`http://markun.cs.shinshu-u.ac.jp/kiso/projects2/proofchecker/mizar/Mizar-J/Miz-tit.htm`

に説明されています。また，PC用のmizarシステムのダウンロード，WEB上で使用できるmizarシステムは

`http://markun.cs.shinshu-u.ac.jp/kiso/projects2/proofchecker/mizar/index-j.html`

にあります。

このテキストでは，mizarで記述された数学定理のデータベースであるライブラリ中の集合論についての，基礎的なarticle

$$TARSKI.miz, XBOOLl0.miz, XBOOLl1.miz$$

について説明します。

1.1　TARSKI.miz

これから,$TARSKI.miz$を読み解いていきます。

まず, 以下の環境部の記述があります。

```
environ
 vocabulary TARSKI;
```

これは, このarticleで, 使われる, 用語 ($vocabulary$) が$TARSKI.voc$というファイルに記述されていることを宣言しています。

次の,$begin$の宣言からこのarticleの内容の記述が始まります。

```
begin

 reserve x,y,z,u,N,M,X,Y,Z for set;
```

reserve はその後に続く, 変数の型がなんであるかを示します。この例では

$$x, y, z, u, N, M, X, Y, Z$$

は任意の集合 *set* になっています。集合論では, 取り扱う対象は集合か, その要素ですが, 集合も要素も, 形式上は区別されません。ですから,*set* というのは, 任意の対象を意味しています。

1.1.1　外延性の公理

(２つの集合が等しいことの定義)
先ず以下の記述があります。

```
theorem
 (for x holds x in X iff x in Y) implies X = Y;
```

記号論理で書けば

$$((\forall x)(x \in X \Leftrightarrow x \in Y)) \Rightarrow X = Y$$

です。
これは, 外延性の公理として知られるものです。
任意の x に対して,

$$x \in X \Leftrightarrow x \in Y$$

であるならば, ２つの集合 X, Y は等しい ($X = Y$ である) ことを主張しています。

1.1.2 非順序対の定義

次の記述は,x 一つだけからなる集合 $\{x\}$ と x, y という二つの要素をもつ集合 $\{y, z\}$ の定義です。

```
definition let y; func { y } means
 x in it iff x = y;
 correctness;
 let z; func { y, z } means
  x in it iff x = y or x = z;
 correctness;
 commutativity;
end;
```

任意の y に対して $\{y\}$ とは, 任意の x について

$$x \in \{y\} \Leftrightarrow x = y$$

を充たす集合であり,

任意の y, z に対して $\{y, z\}$ とは, 任意の x について

$$x \in \{y, z\} \Leftrightarrow x = y \ or \ x = z$$

を充たす集合であること。また *commutativity*(可換性) は

$$\{y, z\} = \{z, y\}$$

であることを表しています。

1.1.3 包含関係の定義

次は,「集合 X が集合 Y の部分集合である」あるは「集合 X が集合 Y に含まれる」という述語の定義が書かれています。

```
definition let X,Y;
 pred X c= Y means
  x in X implies x in Y;
 reflexivity;
end;
```

任意の X, Y に対して

$$X \subseteq Y$$

とは, 任意の x について

$$x \in X \Rightarrow x \in Y$$

が成り立つことをいい,$reflexivity$ は任意の X に対してそれ自身

$$X \subseteq X$$

が成り立つことを表しています。

1.1.4　集合の合弁の定義

任意の集合族 (集合の集合)X に対して,X に属する 集合の全ての合弁が,X から $\cup X$ をつくる $functor$(作用) として定義されています。

```
definition let X;
  func union X means
   x in it iff ex Y st x in Y & Y in X;
 correctness;
end;
```

任意の X に対して,$functor$(作用)$\cup X$ とは,X に,
任意の x に対して

$$(x \in \cup X) \Leftrightarrow ((\exists Y)(x \in Y\ and\ Y \in X))$$

となる集合を対応させる作用であること表しています。

1.1.5　正則性の公理

以下は, 証明が省略された定理として, 記述されていますが, 正則性の公理として
知られるものです。

```
theorem
 x in X implies ex Y st Y in X & not ex x st x in X & x in Y;
```

記号論理で書けば, 任意の X に対して

$$(x \in X) \Rightarrow (\exists Y)(Y \in X \ and \ not(\exists x)(x \in X \ and \ x \in Y))$$

を表しています。x が集合 X の要素であれば, 集合 X の要素であって, しかも,x をその要素として含むような Y は存在しないという主張を表しています。

1.1.6 選出の公理 (Fraenkel の公理式)

集合 A と, 関係式 $P[x,y]$ から集合 X を構成する手続きを $scheme$(公理図式) として記述したのものが以下です。

```
scheme Fraenkel { A()-> set, P[set, set] }:
 ex X st for x holds x in X iff ex y st y in A() & P[y,x]
 provided for x,y,z st P[x,y] & P[x,z] holds y = z;
```

任意の A と,

$$(\forall x, y, z)(P[x,y] \ and \ P[x,z] \Rightarrow y = z)$$

が成り立つ関係 P に対して, 集合 X が存在して,

$$(\forall x)(x \in X \Leftrightarrow (\exists y)(y \in \ \ and \ P[y,x]))$$

となることを表しています。

1.1.7 順序対の定義

任意の x, y に対して

$$\{\{x, y\}, \{x\}\}$$

を $[x, y]$ で表すことを x, y から $[x, y]$ を作り出す $functor$(作用）として定義します。

```
definition let x,y;
 func [x,y] equals :Def5:
  { { x,y }, { x } };
```

```
  coherence;
 end;
canceled;
```

1.1.8　集合の等濃度の定義

以下は, 二つの集合 X, Y の間に, 双射 (一対一, かつ, 上への写像）が存在するとき, 「X, Y の濃度が等しい」*equipotent* と定義すること表しています。

```
definition let X,Y;
 pred X,Y are_equipotent means
   ex Z st
      (for x st x in X ex y st y in Y & [x,y] in Z) &
      (for y st y in Y ex x st x in X & [x,y] in Z) &
       for x,y,z,u st [x,y] in Z & [z,u] in Z holds x = z iff y = u;
end;
```

すなわち, 任意の X, Y に対して, これらが, 等濃度である（*equipotent* である）とは

$$
\begin{aligned}
&(\exists Z)\\
&\{(\forall x : x \in X)(\exists y)(y \in Y \ and \ [x,y] \in Z)\\
&\ and \ (\forall y : y \in Y)(\exists x)(x \in X \ and \ [x,y] \in Z)\\
&\ and \ (\forall x,y,z,u : [x,y] \in Z \ and \ [z,u] \in Z)(x = z \Leftrightarrow y = u)\}
\end{aligned}
$$

が成り立つことを言います。

1.1.9　Tarski の公理

以下は, 証明なしの定理として書かれていますが Tarski の公理として知られるものです。集合 N が与えられたときに, N をその要素として含む集合 M が存在して

1. $X \in M$ かつ $Y \subseteq X$ ならば $Y \in M$ となる。

2. $X \in M$ に対して,$Y \subseteq X$ となる Y は全て $Y \in Z$ となる　$Z \in M$ が存在する。

3. $X \subseteq M$ ならば,X と M は等濃度かあるいは, $X \in M$

という条件を充たすという主張です。

```
theorem
   ex M st N in M &
     (for X,Y holds X in M & Y c= X implies Y in M) &
     (for X st X in M ex Z st Z in M & for Y st Y c= X holds Y in Z) &
     (for X holds X c= M implies X,M are_equipotent or X in M);
```

記号論理で書けば,

$$
\begin{aligned}
&(\exists M)\\
&\{N \in M\\
&\ \ and\ (\forall X, Y)((X \in M\ and\ Y \subseteq X) \Rightarrow Y \in M)\\
&\ \ and\ (\forall X : X \in M)(\exists Z : Z \in M)(\forall Y)((Y \subseteq X) \Rightarrow (Y \in Z))\\
&\ \ and\ (\forall X)((X \subseteq M) \Rightarrow (X, M \text{ は等濃度 } or\ X \in M))\}
\end{aligned}
$$

1.2 xbool.0.miz

次に $xbool.0$ を読み解いてみます。
まず, 始めの環境部は

```
environ

 vocabulary TARSKI, BOOLE, ZFMISC_1;
 constructors TARSKI;
 notation TARSKI;
 theorems TARSKI;
 schemes TARSKI;
```

と書かれています。これは,

1. この $xbool.0$ で使う用語 $(vocabulary)$ が $TARSKI.voc, BOOLE, voc, ZFMISC_1.voc$ というファイルに書かれていること。

2. $TARSIKI$
3. この $xbool.0$ で既に証明済みとして引用する定理が $TARSIKI.abs$ というファイルに書かれていること。
4. この $xbool.0$ で引用する公理図式 $schemes$ が $TARSIKI.abs$ というファイルに書かれていること。

を表しています。

```
reserve X, Y, Z, x, y, z for set;
```

では X, Y, Z, x, y, z が任意の集合, すなわち, 任意の変数であることを宣言しています。

以下,$xbool.0$ の本文が書かれています。

1.2.1 分出公理

前述した選出の公理 $scheme\ Fraenkel$ を使って, 任意の集合 A と関係式 $P[x]$ から,A の要素であって, 関係式 $P[x]$ が成り立つ x 全体の集合

$$\{x|x \in A\ and\ P[x]\}$$

を構成するための公理図式 (分出公理) が書かれています。

```
scheme Separation { A()-> set, P[set] } :
   ex X being set st for x being set
   holds x in X iff x in A() & P[x]
 proof
A1: for x,y,z st x = y & P[y] & x = z & P[z] holds y = z;
   consider X such that
A2: for x holds x in X iff ex y st y in A() & y = x
& P[x] from Fraenkel(A1);
  take X;
  let x;
   x in X iff ex y st y in A() & y = x & P[x] by A2;
  hence thesis;
 end;
```

まず,

```
scheme Separation { A()-> set, P[set] } :
```

で, この公理図式の名前が $Separation$ であり, 任意の集合 A と関係式 $P[\cdot]$ についての公理図式であることが宣言されています。次に, その公理図式の中身

```
   ex X being set st for x being set holds x in X iff x in A() & P[x]
```

で,

$$(\exists X)(\forall x)(x \in X \Leftrightarrow x \in A \ and \ P[x])$$

すなわち, 「任意の x について,x が X の要素であることと, x が集合 A の要素であって, かつ, 関係式 $P[x]$ を充たすことが同値である集合 X が存在する」ことが書かれています。以下は, その証明が書かれています。

```
 proof
A1: for x,y,z st x = y & P[y] & x = z & P[z]
holds y = z;
   consider X such that
A2: for x holds x in X iff ex y st y in A() & y = x
& P[x] from Fraenkel(A1);
  take X;
  let x;
   x in X iff ex y st y in A() & y = x & P[x] by A2;
  hence thesis;
 end;
```

$$x = y \ and \ P[y]$$

という関係式を $Q[x, y]$ で置き換えてみると判りやすいでしょう。

$$A1 : (\forall x, y, z)((Q[x, y] \ and \ Q[x, z]) \Rightarrow \ (y = z))$$

すると,$A1$ から公理図式 $Fraenkel$ によって

$$A2 : (\forall x)(x \in X \Leftrightarrow (\exists y)(y \in A \ and \ y = x \ and \ P[x])$$

となる集合 X が少なくとも一つ存在します。次の

```
  take X;
```

は, この条件を充たす集合 X を選択するという意味です。次に続く

```
  let x;
   x in X iff ex y st y in A() & y = x & P[x] by A2;
  hence thesis;
```

は, x を任意にとると,$A2$ により

$$x \in X \Leftrightarrow (\exists y)(y \in A \ and \ y = x \ and \ P[x])$$

が成り立つことを示しています。さらに,

$$(\exists y)(y \in A \ and \ y = x \ and \ P[x]) \Leftrightarrow x \in A \ and \ P[x]$$

ですが, このような, 述語論理の単純な同値変形は $mizar$ システムが自動的に行いますので,
任意の x について

$$x \in X \Leftrightarrow x \in A \ and \ P[x]$$

が示されたことになり, 証明が終了したことを示す。

```
  hence thesis
```

が書かれます。

1.2.2 空集合の定義と一意性

```
definition
  func {} -> set means
:Def1:   not ex x being set st x in it;
  existence
    proof
      consider X;
      consider Y such that
A1:     x in Y iff x in X & contradiction from Separation;
     take Y;
     thus thesis by A1;
    end;
  uniqueness
    proof
     let X,Y;
     assume (not ex x st x in X) & (not ex x st x in Y);
      then x in Y iff x in X;
     hence thesis by TARSKI:2;
    end;
```

ここでは, 空集合を引数のない作用 {} として定義しています。

```
definition
  func {} -> set means
:Def1:   not ex x being set st x in it;
```

がその定義で,

$$not\ (\exists x)(x \in \{\})$$

が成り立つもの, すなわち, それに含まれる要素がない集合として定義されます。この定義が正しいものとされるためには, このような集合の存在と, それが唯一つに定まっている必要があります。

```
  existence
```

に続く記述がその存在を証明しています。
まず, 適当な集合 X を選び, 次に, 分出の公理 $Separation$ によって,

$$A1 : x \in Y \Leftrightarrow x \in X\ and\ contradiction$$

となる集合 Y を作ります。$contradiction$ は記号論理でいう恒偽命題です。この Y を選べば,{} の条件を充たしますので, 証明終了の記述

```
     thus thesis by A1;
```

が書かれます。次に, 一意性の証明が続きます。

```
  uniqueness
```

に続く記述がその証明です。
X, Y を任意の集合として,

```
     assume (not ex x st x in X) & (not ex x st x in Y);
```

によって,X, Y が共に空集合の条件を充たすこと, すなわち,

$$not(\exists x)(x \in X)\ and\ not(\exists x)(x \in Y)$$

を仮定します。

```
 then x in Y iff x in X;
     hence thesis by TARSKI:2;
```

は, その仮定を使うと

$$x \in Y \leftrightarrow x \in X$$

が成り立つことが示されます。最後に $TARSKI$ というファイルの既に証明済みの定理 $TARSKI : 2$

```
theorem
 (for x holds x in X iff x in Y) implies X = Y;
```

$$((\forall x)(x \in X \Leftrightarrow x \in Y)) \Rightarrow X = Y$$

によって

$$X = Y$$

が示され, 証明が終ります。

1.2.3　集合 X, Y の合弁 $X \cup Y$

以下の記述は,X, Y の合弁 $X \cup Y$ を X, Y への作用 ($functor$) として定義しています。

```
 let X,Y be set;
  func X \/ Y -> set means
:Def2:   x in it iff x in X or x in Y;
  existence
   proof
    take union {X,Y};
    let x;
    thus x in union {X,Y} implies x in X or x in Y
     proof
      assume  x in union {X,Y};
       then ex X0 being set st x in X0 & X0 in {X,Y} by TARSKI:def 4;
      hence thesis by TARSKI:def 2;
     end;
    assume x in X or x in Y;
     then consider X0 being set such that
A4:   X0 = X or X0 = Y and
A5:   x in X0;
     X0 in {X,Y} by A4,TARSKI:def 2;
    hence x in union {X,Y} by TARSKI:def 4,A5;
   end;
  uniqueness
   proof
    let A1, A2 be set such that
A6:  x in A1 iff x in X or x in Y and
```

```
A7:   x in A2 iff x in X or x in Y;
      now
       let x;
        x in A1 iff x in X or x in Y by A6;
       hence x in A1 iff x in A2 by A7;
      end;
     hence A1 = A2 by TARSKI:2;
    end;
   commutativity;
   idempotence;
```

文の冒頭の

```
 let X,Y be set;
  func X \/ Y -> set means
:Def2:   x in it iff x in X or x in Y;
```

は X, Y の合弁 $X \cup Y$ が

$$x \in X \cup Y \Leftrightarrow (x \in X \ or \ x \in Y)$$

を充たすものとして定義されます。$X \cup Y$ は

```
X \/ Y
```

で表されています。

```
  existence
```

から存在性の証明が記述されています。
まず, 既に定義されている

$$union\{X, Y\}$$

について, x を任意にとると,

$$x \in union\{X, Y\} \Rightarrow (x \in X \ or \ x \in Y)$$

が成り立ちます。その証明は

$$x \in union\{X, Y\}$$

を仮定すると, $TARSKI : def4$ に書かれた $union$ の定義によって

$$(\exists X0)(x \in X0 \ and \ X0 \in \{X, Y\}$$

が成り立ち, さらに $TARSKI:def4$ に書かれた $\{X,Y\}$ の定義によって,

$$X0 = X\ or\ X0 = Y$$

が成り立つので, 結局

$$x \in X\ or\ x \in Y$$

が示され証明終了です。
　逆に,

$$x \in X\ or\ x \in Y$$

を仮定すると, この仮定から,

$$A4 : X0 = X\ or\ X0 = Y$$

でかつ,

$$A5 : x \in X0;$$

となる $X0$ を選ぶことができます。このことから, この $X0$ については,$TARSKI:def2$ に書かれた $\{X,Y\}$ によって,

$$X0 \in \{X,Y\}$$

であり,$TARSKI:def4$ に書かれた $union\{X,Y\}$, 上記の $A4$, $A5$ によって,

$$x \in union\{X,Y\}$$

となり, 存在性の証明が終了します。
　一意性の証明は

```
uniqueness
```

の後に書かれています。
　集合 $A1$, $A2$ が

$$A6 : x \in A1 \Leftrightarrow (x \in X\ or\ x \in Y)$$

かつ

$$A7 : x \in A2 \Leftrightarrow (x \in X\ or\ x \in Y)$$

を充たすものとします。このとき, x を任意にとると,$A6$ により,

$$x \in A1 \Leftrightarrow (x \in X\ or\ x \in Y)$$

であり, これから,$A7$ により,

$$x \in A1 \Leftrightarrow x \in A2$$

となります。従って,$TARSKI:2$ に書かれている定理によって

$$A1 = A2$$

が成り立ち, 一意性の証明が終了します。

```
  commutativity;
  idempotence;
```

は,X, Y の順序を替えた $Y \cup X$ も同じものであり, $X = Y$ の場合の $X \cup X$ も定義されることを示しています。

1.2.4 集合 X, Y の共通部分 $X \cap Y$

以下の記述は,X, Y の共通部分 $X \cap Y$ を X, Y への作用 $(functor)$ として定義しています。

文の冒頭の

```
  func X /\ Y -> set means
:Def3:   x in it iff x in X & x in Y;
```

は X, Y の共通部分 $X \cap Y$ が

$$x \in X \cap Y \Leftrightarrow (x \in X \ and \ x \in Y)$$

を充たすものとして定義されます。$X \cap Y$ は

```
X /\ Y
```

で表されています。

存在性は, 既にこのファイルで証明した分出の公理 $Separation$ から直ぐに言えますので,

```
  existence from Separation;
```

とだけ記述されます。

一意性の証明は

```
  uniqueness
```

の後に書かれています。
　集合 $A1, A2$ が

$$A8 : x \in A1 \Leftrightarrow (x \in X \ and \ x \in Y)$$

かつ

$$A9 : x \in A2 \Leftrightarrow (x \in X \ and \ x \in Y)$$

を充たすものとします。このとき, x を任意にとると,$A8$ により,

$$x \in A1 \Leftrightarrow (x \in X \ and \ x \in Y)$$

であり, これから,$A9$ により,

$$x \in A1 \Leftrightarrow x \in A2$$

となります。従って,$TARSKI:2$ に書かれている定理によって

$$A1 = A2$$

が成り立ち, 一意性の証明が終了します。

```
  commutativity;
  idempotence;
```

は,X, Y の順序を替えた $Y \cap X$ も同じものであり, $X = Y$ の場合の $X \cap X$ も定義されることを示しています。

1.2.5　集合 X から Y の要素を除いた差集合 $X \setminus Y$

　以下の記述は, 集合 X から Y の要素を除いた差集合 $X \setminus Y$ を X, Y への作用 $(functor)$ として定義しています。

```
  func X \ Y -> set means
:Def4:   x in it iff x in X & not x in Y;
  existence from Separation;
  uniqueness
   proof
    let A1, A2 be set such that
A10: x in A1 iff x in X & not x in Y and
A11: x in A2 iff x in X & not x in Y;
     now
```

```
      let x;
       x in A1 iff x in X & not x in Y by A10;
      hence x in A1 iff x in A2 by A11;
     end;
    hence A1 = A2 by TARSKI:2;
   end;
end;
```

文の冒頭の

```
  func X \ Y -> set means
:Def4:   x in it iff x in X & not x in Y;
```

は X, Y の差集合 $X \setminus Y$ が

$$x \in X \setminus Y \Leftrightarrow (x \in X \ and\ not(\ x \in Y))$$

を充たすものとして定義されます。$X \setminus Y$ は

```
X \ Y
```

で表されています。

存在性は, 既にこのファイルで証明した分出の公理 $Separation$ から直ぐに言えますので,

```
  existence from Separation;
```

とだけ記述されます。

一意性の証明は

```
  uniqueness
```

の後に書かれています。

集合 $A1, A2$ が

$$A10 : x \in A1 \Leftrightarrow (x \in X \ andnot(\ x \in Y))$$

かつ

$$A11 : x \in A2 \Leftrightarrow (x \in X \ andnot(\ x \in Y))$$

を充たすものとします。このとき, x を任意にとると,$A10$ により,

$$x \in A1 \Leftrightarrow (x \in X \ andnot(\ x \in Y))$$

であり, これから,$A11$ により,

$$x \in A1 \Leftrightarrow x \in A2$$

となります。従って,$TARSKI:2$ に書かれている定理によって

$$A1 = A2$$

が成り立ち, 一意性の証明が終了します。

1.2.6　空集合

```
definition let X be set;
 attr X is empty means
:Def5:    X = {};
```

この定義は,「X が空 (集合) である」という X の属性を

$$X = \{\}$$

が成り立つこととして定義します。

1.2.7　排他論理和

以下の記述は, 集合 X から Y の要素を除いた差集合 $X \setminus Y$ と集合 Y から X の要素を除いた差集合 $Y \setminus X$ の和である排他論理和の集合を X, Y への作用 $(functor)$ として定義しています。

```
 let Y be set;
 func X \+\ Y -> set equals
:Def6:  (X \ Y) \/ (Y \ X);
 correctness;
 commutativity;
```

文の冒頭の

```
 let Y be set;
 func X \+\ Y -> set equals
:Def6:  (X \ Y) \/ (Y \ X);
```

は X, Y の排他論理和の集合 $X \oplus Y$ が

$$X \setminus Y \cup Y \setminus X$$

と等しいものとして定義されます。$X \oplus Y$ は

```
X \ + \ Y
```

で表されています。

存在性は, 既に,$X \setminus Y$ も,$Y \setminus X$ も示されており, それの和集合

$$X \setminus Y \cup Y \setminus X$$

の存在も自明なので,

```
 correctness;
```

とだけ記述され省略されています。また,

```
 commutativity;
```

は,X と Y を入れ替えても同じであることを示しています。(可換性)

1.2.8 互いに素 (交わらない) 集合

以下は,「集合 X,Y とが交わらない (共通部分がない)」という述語を

$$X \cap Y = \phi$$

が成り立つこととして定義しています。空集合 ϕ は {} で表されています。

```
 pred X misses Y means :Def7:
   X /\ Y = {};
  symmetry;
  antonym X meets Y;
```

```
  symmetry;
```

は,「集合 X,Y とが交わらない」ならば「集合 Y,X とが交わらない」も成り立つことを示しています。(対称性)

```
  antonym X meets Y;
```

は, この「集合 X,Y とが交わらない」と反対の意味をもつ述語が,「集合 X,Y とが交わる」

$$X\ meets\ Y$$

であることを表しています。

1.2.9　真部分集合

以下の記述は, 集合 X が Y の部分集合で, かつ,Y と等しくないこと, すなわち

$$(X \subset Y)\ and\ X \neq Y$$

が成り立つことを「集合 X は集合 Y の真部分集合」であるという述語として定義しています。部分集合の場合 $c =(\subseteq)$ を用いましたので真部分集合には $c <$ を用いています。(数学の教科書では $\subset$ などが使われています。)

```
pred X c< Y means
 X c= Y & X <> Y;
irreflexivity;
```

```
irreflexivity;
```

は,$c =(\subseteq)$ の場合と異なり,

$$X \subset X$$

は成り立たないことを表しています。これは

$$(X \subset X)\ and\ X \neq X$$

を意味しますので成り立ちません。

1.2.10　比較可能な集合

以下の記述は, 集合 X と Y が集合の抱合関係として比較可能なこと, すなわち

$$(X \subseteq Y)\ or\ (Y \subseteq X)$$

が成り立つことを,$are_c = -comparable$ という述語として定義しています。

```
pred X,Y are_c=-comparable means
 X c= Y or Y c= X;
reflexivity;
symmetry;
```

```
reflexivity;
```

は,X と X 自身が比較可能であること ($X \subseteq X$ が成り立っていますので)

```
symmetry;
```

は,X と Y が比較可能であれば, Y と X も比較可能であることを表しています。実際,

$$(X \subseteq Y)\ or\ (Y \subseteq X)$$

なら

$$(Y \subseteq X)\ or\ (X \subseteq Y)$$

が成り立っています。

1.2.11 集合の等号 = の再定義

以下は, 既に $TARSKI:2$ で定義されている, 2 つの集合 X, Y についての等号関係 $X = Y$ を

$$(X \subseteq Y)\ and\ (Y \subseteq X)$$

が成り立つことという述語として再定義しています。

```
 redefine pred X = Y means
  X c= Y & Y c= X;
  compatibility
   proof
    thus X = Y implies X c= Y & Y c= X;
    assume  X c= Y & Y c= X;
     then for x holds x in X iff x in Y by TARSKI:def 3;
    hence X = Y by TARSKI:2;
   end;
end;
```

```
redefine
```

は再定義の宣言です。

```
compatibility
```

以下は, この再定義が,$TARSKI:2$ での定義と矛盾しないこと (替えることができること) の証明です。まず,

$$X = Y \Rightarrow (X \subseteq Y)\ and\ (Y \subseteq = X)$$

は自明です。(左辺の $=$ は $TARSKI:2$ の定義での等号です。) 逆に,

$$(X \subseteq Y)\ and\ (Y \subseteq= X)$$

を仮定すると,$TARSKI:def3$ により,

$$(\forall x)(x \in X \Leftrightarrow x \in Y)$$

従って,$TARSKI:2$ により

$$X = Y$$

が成り立ちます。

1.2.12　排他和の定理

以下は定理

$$(x \in X \oplus Y) \Leftrightarrow not(x \in X \Leftrightarrow x \in Y)$$

の証明です。

```
theorem
  x in X \+\ Y iff not (x in X iff x in Y)
proof
  X \+\ Y = (X \ Y) \/ (Y \ X) by Def6;
  then x in X \+\ Y iff x in X \ Y or x in Y \ X by Def2;
  hence thesis by Def4;
end;
```

定義 $Def6$ により,

$$X \oplus Y = (X \setminus Y) \cup (Y \setminus X)$$

が成り立ち, これから,

$$x \in X \oplus Y \Leftrightarrow x \in (X \setminus Y)\ or\ x \in (Y \setminus X)$$

が成り立ち, 定義 $Def4$ により証明終了です。

1.2.13 定理 $BOOLE'25$

以下は

$$(\forall x)(not(x \in X) \Leftrightarrow (x \in Y \Leftrightarrow x \in Z)) \Leftarrow X = Y \oplus Z$$

の証明です。

```
theorem :: BOOLE'25:
  (for x holds not x in X iff (x in Y iff x in Z))   implies X = Y \+\ Z
proof
  assume A1: not x in X iff (x in Y iff x in Z);
  now let x;
    x in X iff x in Y & not x in Z or x in Z & not x in Y by A1;
    then x in X iff x in Y \ Z or x in Z \ Y by Def4;
    then x in X iff x in (Y \ Z) \/ (Z \ Y) by Def2;
    hence x in X iff x in Y \+\ Z by Def6;
  end;
  hence thesis by TARSKI:2;
end;
```

$$A1 : not(x \in X) \Leftrightarrow (x \in Y \Leftrightarrow x \in Z))$$

を仮定すると, 任意の x について,$A1$ から,

$$(x \in X) \Leftrightarrow ((x \in Y \ and\ not\ x \in Z)\ or\ (x \in Z\ and\ not\ x \in Y))$$

が成り立ち, これと $Def4$ から

$$x \in X \Leftrightarrow (x \in Y \setminus Z\ or\ x \in Z \setminus Y)$$

が成り立ち, さらに,$Def2$ により,

$$x \in X \Leftrightarrow x \in (Y \setminus Z) \cup (Z \setminus Y)$$

が成り立ち, これから $TARSKI : 2$ によって証明終了です。

1.2.14 $cluster$ の宣言

以下は, 集合 $\{\}(\phi)$ は $emty$ であり, $emty$ は集合であること, さらに, 空でない集合 $nonempty$ も集合であることを主張しています。このような概念の包含関係を示すには $cluster$ を用います。

```
definition
 cluster {} -> empty;
  coherence by Def5;
 cluster empty set;
  existence
   proof {} is empty by Def5; hence thesis; end;
 cluster non empty set;
  existence
   proof consider x;
    x in {x} by TARSKI:def 1;
    then {x} <> {} by Def1;
    then {x} is non empty by Def5;
    hence thesis;
   end;
end;
```

```
 cluster empty set;
  existence
   proof {} is empty by Def5; hence thesis; end;
```

の記述で,$empty$ であるものは集合 (set) であると, 概念の包含関係を示し, さらに,$empty$ である対象が少なくとも一つ存在することを証明しています。$Def5$ により, {} がその $empty$ であるものの一つです。

同様に,

```
 cluster non empty set;
  existence
   proof consider x;
    x in {x} by TARSKI:def 1;
    then {x} <> {} by Def1;
    then {x} is non empty by Def5;
    hence thesis;
   end;
end;
```

の記述で,$nonempty$ であるものは集合 (set) であると, 概念の包含関係を示し, さらに,$nonempty$ である対象が少なくとも一つ存在することを証明しています。実際,$TARSKI : def1$ により,

$$x \in \{x\}$$

が成り立ち, これは $Def1$ により,

$$\{x\} \neq \phi$$

従って,$\{x\}$ は,$nonempty$ です。

1.2.15 定理 $BOOLE'1$

以下は, 定理

$$X \text{ と } Y \text{ は交わる} \Leftrightarrow (\exists x)(x \in X \ and \ x \in Y)$$

の証明です。

```
theorem BOOLE'1: :: BOOLE'1:
  X meets Y iff ex x st x in X & x in Y
proof
 hereby assume X meets Y;
  then X /\ Y <> {} by Def7;
  then consider x such that
A1: x in X /\ Y by Def1;
  take x;
  thus x in X & x in Y by Def3,A1;
 end;
 given x such that
A2: x in X & x in Y;
  x in X /\ Y by Def3,A2;
  then X /\ Y <> {} by Def1;
 hence thesis by Def7;
end;
```

証明は,

$$X \text{ と } Y \text{ は交わる}$$

を仮定すると,$Def7$ から,

$$X \cap Y \neq \phi$$

よって,$Def1$ により,

$$A1 : x \in X \cap Y$$

となる x が存在し, このような x を選れべば, $A1$ と $Def1$ により,

$$x \in X \ and \ x \in Y$$

となり,

$$(\exists x)(x \in X \ and \ x \in Y)$$

が示されます。逆に,

$$(\exists x)(x \in X \ and \ x \in Y)$$

を仮定すれば,

$$A2 : x \in X \ and \ x \in Y$$

となる x を選ぶことができ,$Def3$, $A2$ から

$$x \in X \cap Y$$

が成り立ち従って,$Def1$ により,

$$X \cap Y \neq \phi$$

が示され, これと $Def7$ から証明終了です。

1.2.16 定理 $BOOLE'2$

以下は, 定理

$$X \text{ と } Y \text{ は交わる} \Leftrightarrow (\exists x)(x \in X \cap Y)$$

の証明です。

```
theorem :: BOOLE'2:
  X meets Y iff ex x st x in X /\ Y
proof
 hereby assume X meets Y;
  then X /\ Y <> {} by Def7;
  then consider x such that
A1: x in X /\ Y by Def1;
  take x;
  thus x in X /\ Y by A1;
 end;
 given x such that
A2: x in X /\ Y;
  X /\ Y <> {} by A2,Def1;
 hence thesis by Def7;
end;
```

証明は,

$$X \text{ と } Y \text{ は交わる}$$

を仮定すると,$Def7$ から,

$$X \cap Y \neq \phi$$

よって,$Def1$ により,

$$A1 : x \in X \cap Y$$

となる x が存在し, このような x を選れべば, $A1$ により,

$$x \in X \cap Y$$

となり,

$$(\exists x)(x \in X \cap Y)$$

が示されます。逆に,

$$(\exists x)(x \in X \cap Y)$$

を仮定すれば,

$$A2 : x \in X \cap Y$$

となる x を選ぶことができ,$Def2, A1$ から

$$X \cap Y \neq \phi$$

が成り立ち従って, これと $Def7$ から証明終了です。

1.2.17 定理 $SYSREL'1$

以下は

$$((X \text{ と } Y \text{ が交わない})\ and\ (x \in X \cap Y))$$
$$\Rightarrow$$
$$((x \in X\ and\ not(x \in Y))\ or\ (x \in Y\ and\ not(x \in X)))$$

を表す定理です。

```
theorem :: SYSREL'1:
  X misses Y & x in X \/ Y implies
    ((x in X & not x in Y) or (x in Y & not x in X))
      by Def2,BOOLE'1;
```

1.2.18 *scheme Extensionality*

以下は, 集合 X, Y と関係式 P について

$$(\forall x)(x \in X1 \Leftrightarrow P[x]) \ and \ (\forall x)(x \in X2 \Leftrightarrow P[x]))$$

が成り立つとき,

$$X = Y$$

が成り立つことを表す $scheme$(公理図式) です。

```
scheme Extensionality { X,Y() -> set, P[set] } :
  X() = Y() provided
A1:  for x holds x in X() iff P[x] and
A2:  for x holds x in Y() iff P[x]
  proof
A3:  x in X() implies x in Y()
     proof assume x in X(); then P[x] by A1; hence x in Y() by A2; end;
    x in Y() implies x in X()
     proof assume x in Y(); then P[x] by A2; hence x in X() by A1; end;
   hence thesis by TARSKI:2,A3;
  end;
```

証明は,

$$A1 : (\forall x)(x \in X1 \Leftrightarrow P[x])$$

$$A2 : (\forall x)(x \in X2 \Leftrightarrow P[x])$$

を仮定すれば,

$$x \in X \Rightarrow P[x]$$
$$P[x] \Rightarrow x \in Y$$

によって,

$$A3 : x \in X \Rightarrow x \in Y$$

が成り立ち, 同様に,

$$x \in Y \Rightarrow P[x]$$
$$P[x] \Rightarrow x \in X$$

によって,

$$x \in Y \Rightarrow x \in X$$

が成り立つことが示され, これと,$TARSKI : 2$ 及び $A3$ から, 証明終了です。

1.2.19 *scheme SetEq*

以下は,

$(\forall X1, X2 : (\forall x)(x \in X1 \Leftrightarrow P[x]) \; and \; (\forall x)(x \in X2 \Leftrightarrow P[x]))(X1 = X2)$

を表す公理図式です。

```
scheme SetEq { P[set] } :
  for X1,X2 being set st
   (for x being set holds x in X1 iff P[x]) &
   (for x being set holds x in X2 iff P[x]) holds X1 = X2
proof let X1,X2 be set such that
  A1: for x being set holds x in X1 iff P[x] and
  A2: for x being set holds x in X2 iff P[x];
  thus thesis from Extensionality(A1,A2);
end;
```

証明は,

$$A1 : (\forall x)(x \in X1 \Leftrightarrow P[x])$$

$$A2 : (\forall x)(x \in X2 \Leftrightarrow P[x])$$

を仮定すれば, 関係式 $A1, A2$ に公理図式 $Extensionality(A1, A2)$ が適用できることによって示されます。

1.3 xboole1.miz

$TARSKI.miz$ と $xboole0.miz$ は集合論 ($TARSKI$ の公理系) についての公理と集合演算の定義が記述されていました。$xboole1.miz$ ではそれらを用いて，集合の基礎的な定理が展開されています。ここではその一部を読み解くことにします。

ファイルの冒頭には，既に説明しましたように,『環境部』があります。

```
:: Boolean Properties of Sets - Theorems
::   Library Committee
::
:: Received April 08, 2002
:: Copyright (c) 2002 Association of Mizar Users
```

```
environ

 vocabulary BOOLE, ZFMISC_1;
 constructors TARSKI, XBOOLE_0;
 requirements BOOLE;
 notation TARSKI, XBOOLE_0;
 clusters XBOOLE_0;
 definitions TARSKI, XBOOLE_0;
 theorems TARSKI, XBOOLE_0;

begin

 reserve x,A,B,X,X',Y,Y',Z,V for set;
```

ここで使う記号法 $(vocabulary)$ は $BOOLE.voc$ と $ZFMISC\_1.voc$ に登録されています。

また，その記号法を用いた述語や用語 $notation$ は，$TARSKI, XBOOLE\_0$ に記述されています。

それらの用語や記号を用いた定義 $definitions$ は $TARSKI, XBOOLE\_0$ から引用され，定理 $theorems$ も同様に $TARSKI, XBOOLE\_0$ から引用されます。

以下，定理の記述を読んで行きます。

1.3.1 BOOLE'1

```
BOOLE'1: X meets Y iff ex x st x in X & x in Y by XBOOLE_0:3;
```

これは，殆ど，説明の必要がないでしょうが $XBOOLE\_0:3$ から，

$$X\ meets\ Y \Leftrightarrow (\exists x)(x \in X\ and\ x \in Y)$$

1.3.2 BOOLE'2

```
X meets Y iff ex x st x in X /\ Y by XBOOLE_0:4;
```

これも，殆ど，説明の必要がないでしょうが
$XBOOLE\_0:4$ から，

$$BOOLE'2 : X\ meets\ Y \Leftrightarrow (\exists x)(x \in X \cap\ Y)$$

1.3.3 theorem BOOLE'29: :: BOOLE'29:

```
theorem BOOLE'29: :: BOOLE'29:
  X c= Y & Y c= Z implies X c= Z
proof
  assume that A1: X c= Y and
              A2: Y c= Z;
   let x; assume x in X; then x in Y by A1,TARSKI:def 3;
   hence thesis by A2,TARSKI:def 3;
end;
```

これは，以下の通りです。

$$X \subseteq Y \ and \ Y \subseteq Z \Rightarrow X \subseteq Z$$

証明

$A1 : X \subseteq Y$
$A2 : Y \subseteq Z$
を仮定して x を任意にとり，さらに
$x \in X$ を仮定すると, $A1$ と $TARSKI : def3$ から
$x \in Y$

故に $A2, TARSKI : def3$ から

$$X \subseteq Y \ and \ Y \subseteq Z \Rightarrow X \subseteq Z$$

証明終了

1.3.4 theorem BOOLE'27: :: BOOLE'27:

```
theorem BOOLE'27: :: BOOLE'27:
  {} c= X
proof let x;
  thus thesis by XBOOLE_0:def 1;
end;
```

これは以下の通りです。

$$\{\} \subseteq X$$

証明

$$x を任意にとると XBOOLE\_0 : def1 により,$$
$$x \in \{\} \Rightarrow x \in X$$

証明終了

1.3.5 theorem BOOLE'30: :: BOOLE'30:

```
theorem BOOLE'30: :: BOOLE'30:
  X c= {} implies X = {}
proof
  assume X c= {};
  hence X c= {} & {} c= X by BOOLE'27;
end;
```

これは以下の通りです。

$$X \subseteq \{\} \Rightarrow X = \{\}$$

証明

$$X \subseteq \{\} を仮定すると BOOLE'27 により,$$
$$X \subseteq \{\}\ and\ \{\} \subseteq X$$

証明終了

1.3.6 theorem BOOLE'64: :: BOOLE'64:

```
theorem BOOLE'64: :: BOOLE'64:
  (X \/ Y) \/ Z = X \/ (Y \/ Z)
proof
 thus (X \/ Y) \/ Z c= X \/ (Y \/ Z)
   proof let x;
      assume x in (X \/ Y) \/ Z;
```

```
      then x in X \/ Y or x in Z by XBOOLE_0:def 2;
      then x in X or x in Y or x in Z by XBOOLE_0:def 2;
      then x in X or x in Y \/ Z by XBOOLE_0:def 2;
      hence thesis by XBOOLE_0:def 2;
   end;
  let x;
  assume x in X \/ (Y \/ Z);
  then x in X or x in Y \/ Z by XBOOLE_0:def 2;
  then x in X or x in Y or x in Z by XBOOLE_0:def 2;
  then x in X \/ Y or x in Z by XBOOLE_0:def 2;
  hence thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$(X \cup Y) \cup Z = X \cup (Y \cup Z)$$

証明

まず, (1) $(X \cup Y) \cup Z \subseteq X \cup (Y \cup Z)$

(1) の証明

x を任意にとり,

$x \in (X \cup Y) \cup Z$ を仮定すると

$\therefore XBOOLE\_0 : def2$ により, $x \in X \cup Y \ or \ x \in Z$

$\therefore XBOOLE\_0 : def2$ により, $x \in X \ or \ x \in Y \ or \ x \in Z$

$\therefore XBOOLE\_0 : def2$ により, $x \in X \ or \ x \in Y \cup Z$

故に $XBOOLE\_0 : def2$ により,

$(X \cup Y) \cup Z \subseteq X \cup (Y \cup Z)$

(1) の証明終了

次に x を任意にとり, $x \in X \cup (Y \cup Z)$ を仮定すると ：

$XBOOLE\_0 : def2$ により, $x \in X \ or \ x \in Y \cup Z$

$\therefore XBOOLE\_0 : def2$ により, $x \in X \ or \ x \in Y \ or \ x \in Z$

$\therefore XBOOLE\_0 : def2$ により, $x \in X \cup Y \ or \ x \in Z$

故に $XBOOLE\_0 : def2$ により,

$X \cup (Y \cup Z) \quad \subseteq \quad (X \cup Y) \cup Z$

よってこれと (1) から

$$(X \cup Y) \cup Z = X \cup (Y \cup Z)$$

証明終了

1.3.7 theorem :: SYSREL'2:

```
theorem :: SYSREL'2:
  (X \/ Y) \/ Z = (X \/ Z) \/ (Y \/ Z)
proof
  (X \/ Y) \/ Z = X \/ ((Z \/ Z) \/ Y) by BOOLE'64
             .= X \/ (Z \/ (Z \/ Y)) by BOOLE'64
             .= (X \/ Z) \/ (Y \/ Z) by BOOLE'64;
  hence thesis;
end;
```

これは以下の通りです。

$$(X \cup Y) \cup Z = (X \cup Z) \cup (Y \cup Z)$$

証明

$$\begin{aligned} (X \cup Y) \cup Z &= X \cup ((Z \cup Z) \cup Y) \quad (BOOLE'64 \text{ による}) \\ &= X \cup (Z \cup (Z \cup Y)) \quad (BOOLE'64 \text{ による}) \\ &= (X \cup Z) \cup (Y \cup Z) \quad (BOOLE'64 \text{ による}) \end{aligned}$$

よって

$$(X \cup Y) \cup Z = (X \cup Z) \cup (Y \cup Z)$$

証明終了

1.3.8 theorem :: SYSREL'3:

```
theorem :: SYSREL'3:
  X \/ (X \/ Y) = X \/ Y
proof
  X \/ (X \/ Y) = (X \/ X) \/ Y by BOOLE'64
                .= X \/ Y;
  hence thesis;
end;
```

これは以下の通りです。

$$X \cup (X \cup Y) = X \cup Y$$

証明

$$\begin{aligned} X \cup (X \cup Y) &= (X \cup X) \cup Y \quad (BOOLE'64\ \text{による}) \\ &= X \cup Y \end{aligned}$$

故に

$$X \cup (X \cup Y) = X \cup Y$$

証明終了

1.3.9 theorem BOOLE'31: :: BOOLE'31:

```
theorem BOOLE'31: :: BOOLE'31:
  X c= X \/ Y
proof
  let x;
  thus thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$X \subseteq X \cup Y$$

証明

$$\begin{aligned} &x\ \text{を任意にとると},\ XBOOLE\_0 : def2\ \text{により}, \\ &x \in X \Rightarrow x \in X\ or\ x \in Y \end{aligned}$$

証明終了

1.3.10 theorem BOOLE'32: :: BOOLE'32:

```
theorem BOOLE'32: :: BOOLE'32:
  X c= Z & Y c= Z implies X \/ Y c= Z
proof
```

```
  assume A1: X c= Z & Y c= Z;
  let x;
  assume x in X \/ Y;
  then x in X or x in Y by XBOOLE_0:def 2;
  hence thesis by  A1,TARSKI:def 3;
end;
```

これは以下の通りです。

$$X \subseteq Z \ and \ Y \subseteq Z \Rightarrow X \cup Y \subseteq Z$$

証明

$A1 : X \subseteq Z \ and \ Y \subseteq Z$
を仮定し, x を任意にとり，さらに
$x \in X \cup Y$ 次を仮定すると
$XBOOLE\_0 : def2$ により, $x \in X \ or \ x \in Y$

故に $A1, TARSKI : def3$ から

$$X \subseteq Z \ and \ Y \subseteq Z \Rightarrow X \cup Y \subseteq Z$$

証明終了

1.3.11 theorem BOOLE'33: :: BOOLE'33:

```
theorem BOOLE'33: :: BOOLE'33:
  X c= Y implies X \/ Z c= Y \/ Z
proof
  assume A1: X c= Y;
  let x; assume x in X \/ Z;
  then x in X or x in Z by XBOOLE_0:def 2;
  then x in Y or x in Z by A1,TARSKI:def 3;
  hence thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$X \subseteq Y \Rightarrow X \cup Z \subseteq Y \cup Z$$

証明

$$A1 : X \subseteq Y \text{ を仮定し}$$
$$x \text{ を任意にとり，さらに}$$
$$x \in X \cup Z \text{ を仮定すると}$$
$$XBOOLE\_0 : def2 \text{ により}, x \in X \ or \ x \in Z$$
$$\therefore A1, TARSKI : def3 \text{ により } x \in Y \ or \ x \in Z$$

故に $XBOOLE\_0 : def2$ から

$$X \subseteq Y \Rightarrow X \cup Z \subseteq Y \cup Z$$

証明終了

1.3.12 theorem :: AMI_5'2 :

```
theorem :: AMI_5'2:
  X c= Y implies X c= Z \/ Y
 proof
  assume X c= Y; then
A1: Z \/ X c= Z \/ Y by BOOLE'33;
  X c= Z \/ X by BOOLE'31;
  hence X c= Z \/ Y by A1,BOOLE'29;
 end;
```

これは以下の通りです。

$$X \subseteq Y \Rightarrow X \subseteq Z \cup Y$$

証明

$$X \subseteq Y \text{ を仮定すると } BOOLE'33 \text{ から}$$
$$A1 : Z \cup X \subseteq Z \cup Y$$
$$BOOLE'31 \text{ から } X \subseteq Z \cup X$$
$$\text{よっ て } A1, BOOLE'29 \text{ から } X \subseteq Z \cup Y$$

証明終了

1.3.13 theorem :: SETWISEO'7:

```
theorem :: SETWISEO'7:
  X \/ Y c= Z implies X c= Z
proof
  X c= X \/ Y by BOOLE'31;
  hence thesis by BOOLE'29;
end;
```

これは以下の通りです。

$$X \cup Y \subseteq Z \Rightarrow X \subseteq Z$$

証明

$$\begin{aligned}&BOOLE'31\text{ から}, X \subseteq X \cup Y\\&\text{故に }BOOLE'29\text{ から}\\&X \cup Y \subseteq Z \Rightarrow X \subseteq Z\end{aligned}$$

証明終了

1.3.14 theorem BOOLE'35: :: BOOLE'35:

```
theorem BOOLE'35: :: BOOLE'35:
  X c= Y implies X \/ Y = Y
proof
  assume A1: X c= Y;
   thus X \/ Y c= Y
     proof
       let x; assume x in X \/ Y; then
       x in X or x in Y by XBOOLE_0:def 2;
       hence thesis by A1,TARSKI:def 3;
     end;
  let x;
  thus thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$X \subseteq Y \Rightarrow X \cup Y = Y$$

証明

$A1 : X \subseteq Y$ を仮定すると
まず (1) $X \cup Y \subseteq Y$
(1) の証明
x を任意にとり,
$x \in X \cup Y$ を仮定すると,
$XBOOLE\_0 : def2$ から
$x \in X or x \in Y$
故に $A1, TARSKI : def3$ から
$X \cup Y \subseteq Y$
(1) の証明終了
つぎに, x を任意にとると,
$XBOOLE\_0 : def2$ により, $x \in Y \Rightarrow x \in X \cup Y$

ゆえに,

$$X \subseteq Y \Rightarrow X \cup Y = Y$$

証明終了

1.3.15 theorem :: BOOLE'34:

```
theorem :: BOOLE'34:
  X c= Y & Z c= V implies X \/ Z c= Y \/ V
proof
  assume A1: X c= Y;
  assume A2: Z c= V;
  let x; assume x in X \/ Z;
  then x in X or x in Z by XBOOLE_0:def 2;
  then x in Y or x in V by A1,A2,TARSKI:def 3;
  hence thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$X \subseteq Y and Z \subseteq V \Rightarrow X \cup Z \subseteq Y \cup V$$

証明

$$A1 : X \subseteq Y$$
$$A2 : Z \subseteq V$$
を仮定し, x を任意にとり，さらに
$x \in X \cup Z$ を仮定すると
$XBOOLE\_0 : def2$ により, $x \in X\ or\ x \in Z$
$\therefore A1, A2, TARSKI : def3$ から $x \in Y\ or\ x \in V$
$XBOOLE\_0 : def2$ から $X \cup Z \subseteq Y \cup V$

故に
$$X \subseteq Y and Z \subseteq V \Rightarrow X \cup Z \subseteq Y \cup V$$

証明終了

1.3.16 theorem :: BOOLE'56:

```
theorem :: BOOLE'56:
  X = Y \/ Z iff Y c= X & Z c= X & for V st Y c= V & Z c= V holds X c= V
proof
  thus X = Y \/ Z implies
    Y c= X & Z c= X & for V st Y c= V & Z c= V holds X c= V
      by BOOLE'31,BOOLE'32;
  assume that A1: Y c= X and
              A2: Z c= X and
              A3: Y c= V & Z c= V implies X c= V;
  Y c= Y \/ Z & Z c= Y \/ Z by BOOLE'31;
  hence X c= Y \/ Z by A3;
  thus Y \/ Z c= X by A1,A2,BOOLE'32;
end;
```

これは以下の通りです。

$$X = Y \cup Z \Leftrightarrow Y \subseteq X\ and\ Z \subseteq X\ and\ (\forall V : Y \subseteq V\ and\ Z \subseteq V)(X \subseteq V)$$

証明

まず, $BOOLE'31, BOOLE'32$ から

$X = Y \cup Z \Rightarrow Y \subseteq X\ and\ Z \subseteq X\ and\ (\forall V : Y \subseteq V\ and\ Z \subseteq V)(X \subseteq V)$

次に

$A1 : Y \subseteq X$

$A2 : Z \subseteq X$

$A3 : Y \subseteq V\ and\ Z \subseteq V \Rightarrow X \subseteq V$

を仮定すると, $BOOLE'31$ から

$Y \subseteq Y \cup Z\ and\ Z \subseteq Y \cup Z$

従って $A3$ から, $X \subseteq Y \cup Z$

よって, $A1, A2, BOOLE'32$ から $Y \cup Z \subseteq X$

故に

$X = Y \cup Z \Leftrightarrow Y \subseteq X\ and\ Z \subseteq X\ and\ (\forall V : Y \subseteq V\ and\ Z \subseteq V)(X \subseteq V)$

証明終了

1.3.17 theorem :: BOOLE'59:

```
theorem :: BOOLE'59:
  X \/ Y = {} implies X = {}
proof
  assume
A1: X \/ Y = {};
  not ex x st x in X
  proof
    given x such that A3:x in X;
    x in X \/ Y by XBOOLE_0:def 2,A3;
    hence contradiction by A1,XBOOLE_0:def 1;
  end;
  hence thesis by XBOOLE_0:def 1;
end;
```

これは以下の通りです。

$$X \cup Y = \{\} \Rightarrow X = \{\}$$

証明

$A1 : X \cup Y = \{\}$ を仮定すると
(1)　$not(\exists x)(x \in X)$
(1) の証明
$A3 : x \in X$ となる x を選ぶと,
$XBOOLE\_0 : def2$ と $A3$ から $x \in X \cup Y$
これは $A1$ と $XBOOLE\_0 : def1$ から矛盾
(1) の証明終了

故に $XBOOLE\_0 : def1$ により

$$X \cup Y = \{\} \Rightarrow X = \{\}$$

証明終了

1.3.18 theorem BOOLE'67: :: BOOLE'67:

```
theorem BOOLE'67: :: BOOLE'67:
  (X /\ Y) /\ Z = X /\ (Y /\ Z)
proof
  thus (X /\ Y) /\ Z c= X /\ (Y /\ Z)
  proof let x;
    assume x in (X /\ Y) /\ Z;
    then x in X /\ Y & x in Z by XBOOLE_0:def 3;
    then x in X & x in Y & x in Z by XBOOLE_0:def 3;
    then x in X & x in Y /\ Z by XBOOLE_0:def 3;
    hence thesis by XBOOLE_0:def 3;
  end;
  let x;
  assume x in X /\ (Y /\ Z);
  then x in X & x in Y /\ Z by XBOOLE_0:def 3;
  then x in X & x in Y & x in Z by XBOOLE_0:def 3;
  then x in X /\ Y & x in Z by XBOOLE_0:def 3;
```

```
  hence thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$(X \cap Y) \cap Z = X \cap (Y \cap Z)$$

証明

まず, (1) $(X \cap Y) \cap Z \subseteq X \cap (Y \cap Z)$

(1) の証明

x を任意にとり, $x \in (X \cap Y) \cap Z$ 仮定すると

$XBOOLE\_0 : def3$ から, $x \in X \cap Y\ and\ x \in Z$

$XBOOLE\_0 : def3$ から, $x \in X\ and\ x \in Y\ and\ x \in Z$

$XBOOLE\_0 : def3$ から, $x \in X\ and\ x \in Y \cap Z$

故に $XBOOLE\_0 : def3$ から

$(X \cap Y) \cap Z \subseteq X \cap (Y \cap Z)$

(1) の証明終了

次に x を任意にとり, $x \in X \cap (Y \cap Z)$ を仮定すると

$XBOOLE\_0 : def3$ から, $x \in X\ and\ x \in Y \cap Z$

$\therefore XBOOLE\_0 : def3$ から, $x \in X\ and\ x \in Y\ and\ x \in Z$

$\therefore XBOOLE\_0 : def3$ から, $x \in X \cap Y\ and\ x \in Z$

故に $XBOOLE\_0 : def3$ から,

$X \cap (Y \cap Z) \subseteq (X \cap Y) \cap Z$

よって,

$$(X \cap Y) \cap Z = X \cap (Y \cap Z)$$

証明終了

1.3.19 theorem BOOLE'37: :: BOOLE'37:

```
theorem BOOLE'37: :: BOOLE'37:
  X /\ Y c= X
proof
  let x;
  thus thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$X \cap Y \subseteq X$$

証明

$$\begin{aligned}&x\text{ を任意にとると}, XBOOLE\_0 : def3\text{ により}\\&x \in X \cap Y \Rightarrow x \in X\end{aligned}$$

証明終了

1.3.20 theorem :: SYSREL'4:

```
theorem :: SYSREL'4:
  X c= Y /\ Z implies X c= Y
proof Y /\ Z c= Y by BOOLE'37;
  hence thesis by BOOLE'29;
end;
```

これは以下の通りです。

$$X \subseteq Y \cap Z \Rightarrow X \subseteq Y$$

証明

$$\begin{aligned}&BOOLE'37\text{ により}, Y \cap Z \subseteq Y\\&\text{故に } BOOLE'29\text{ により}\\&X \subseteq Y \cap Z \Rightarrow X \subseteq Y\end{aligned}$$

証明終了

1.3.21 theorem BOOLE'39: :: BOOLE'39:

```
theorem BOOLE'39: :: BOOLE'39:
  Z c= X & Z c= Y implies Z c= X /\ Y
proof
  assume A1: Z c= X & Z c= Y;
  let x;
```

```
  assume x in Z;
  then x in X & x in Y by A1,TARSKI:def 3;
  hence thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$Z \subseteq X \ and \ Z \subseteq Y \Rightarrow Z \subseteq X \cap Y$$

証明

$A1 : Z \subseteq X \ and \ Z \subseteq Y$ を仮定し, x を任意にとり, さらに,
$x \in Z$ を仮定すると
$A1, TARSKI : def3$ から, $x \in X \ and \ x \in Y$
故に $XBOOLE\_0 : def3$ から
$Z \subseteq X \ and \ Z \subseteq Y \Rightarrow Z \subseteq X \cap Y$

証明終了

1.3.22 theorem :: BOOLE'57:

```
theorem :: BOOLE'57:
  X = Y /\ Z iff X c= Y & X c= Z & for V st V c= Y & V c= Z holds V c= X
proof
  thus X = Y /\ Z implies
    X c= Y & X c= Z & for V st V c= Y & V c= Z holds V c= X
      by BOOLE'37,BOOLE'39;
  assume that A1: X c= Y and
              A2: X c= Z and
              A3: V c= Y & V c= Z implies V c= X;
  thus X c= Y /\ Z by BOOLE'39,A1,A2;
  Y /\ Z c= Y & Y /\ Z c= Z implies Y /\ Z c= X by A3;
  hence Y /\ Z c= X by BOOLE'37;
end;
```

これは以下の通りです。

$$X = Y \cap Z \Leftrightarrow X \subseteq Y \ and \ X \subseteq Z \ and \ (\forall V : V \subseteq Y \ and \ V \subseteq Z)(V \subseteq X)$$

証明

まず, $BOOLE'37, BOOLE'39$ から
$X = Y \cap Z \Rightarrow$
$X \subseteq Y\ and\ X \subseteq Z\ and\ ((\forall V : V \subseteq Y\ and\ V \subseteq Z) \Leftrightarrow (V \subseteq X))$
逆に
$A1 : X \subseteq Y$
$A2 : X \subseteq Z$
$A3 : V \subseteq Y\ and\ V \subseteq Z \Rightarrow V \subseteq X$
を仮定すると
$BOOLE'39, A1, A2$ により, $X \subseteq Y \cap Z$
$A3$ から
$Y \cap Z \subseteq Y\ and\ Y \cap Z \subseteq Z \Rightarrow Y \cap Z \subseteq X$
従って $BOOLE'37$ から $Y \cap Z \subseteq X$

証明終了

1.3.23 theorem :: BOOLE'68:

```
theorem :: BOOLE'68:
  X /\ (X \/ Y) = X
proof
  thus X /\ (X \/ Y) c= X
  proof let x;
    thus thesis by XBOOLE_0:def 3;
  end;
  let x;
  assume x in X;
  then x in X & x in X \/ Y by XBOOLE_0:def 2;
  hence thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$X \cap (X \cup Y) = X$$

証明

先ず, (1) $X \cap (X \cup Y) \subseteq X$

(1) の証明

x を任意にとると, $XBOOLE\_0 : def3$ により

$x \in X \cap (X \cup Y) \Rightarrow x \in X$

よ っ て $X \cap (X \cup Y) \subseteq X$

(1) の証明終了

x を任意にとり, $x \in X$ を仮定すると

$XBOOLE\_0 : def2$ により $x \in X$ and $x \in X \cup Y$

故に $XBOOLE\_0 : def3$ から

$X \subseteq X \cap (X \cup Y)$

証明終了

1.3.24 theorem BOOLE'69: :: BOOLE'69:

```
theorem BOOLE’69: :: BOOLE’69:
  X \/ (X /\ Y) = X
proof
  thus X \/ (X /\ Y) c= X
  proof let x;
    assume x in X \/ (X /\ Y);
    then x in X or x in X /\ Y by XBOOLE_0:def 2;
    hence thesis by XBOOLE_0:def 3;
  end;
  let x;
  thus thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$X \cup (X \cap Y) = X$$

証明

(1) $X \cup (X \cap Y) \subseteq X$

(1) の証明

x を任意にとり, $x \in X \cup (X \cap Y)$ を仮定すると
$XBOOLE\_0 : def2$ から $x \in X$ or $x \in X \cap Y$
故に $XBOOLE\_0 : def3$ から
$X \cup (X \cap Y) \subseteq X$
(1)　の証明終了
x を任意にとると $XBOOLE\_0 : def2$ から
$x \in X \Rightarrow x \in X \cup (X \cap Y)$

よって

$$X \cup (X \cap Y) = X$$

証明終了

1.3.25　theorem BOOLE'70: :: BOOLE'70:

```
theorem BOOLE'70: :: BOOLE'70:
  X /\ (Y \/ Z) = X /\ Y \/ X /\ Z
proof
  thus X /\ (Y \/ Z) c= X /\ Y \/ X /\ Z
  proof let x;
    assume x in X /\ (Y \/ Z);
    then x in X & x in Y \/ Z by XBOOLE_0:def 3;
    then x in X & (x in Y or x in Z) by XBOOLE_0:def 2;
    then x in X /\ Y or x in X /\ Z by XBOOLE_0:def 3;
    hence thesis by XBOOLE_0:def 2;
  end;
  let x;
  assume x in X /\ Y \/ X /\ Z;
   then x in X /\ Y or x in X /\ Z by XBOOLE_0:def 2;
   then x in X & x in Y or x in X & x in Z by XBOOLE_0:def 3;
   then x in X & x in Y \/ Z by XBOOLE_0:def 2;
  hence thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$X \cap (Y \cup Z) = X \cap Y \cup X \cap Z$$

証明

先ず次が成り立つ

(1) $X \cap (Y \cup Z) \subseteq X \cap Y \cup X \cap Z$

(1) の証明

x を任意にとり $x \in X \cap (Y \cup Z)$ を仮定すると

$XBOOLE\_0 : def3$ から, $x \in X\ and\ x \in Y \cup Z$

$XBOOLE\_0 : def2$ から, $x \in X\ and\ (x \in Y\ or\ x \in Z)$

$XBOOLE\_0 : def3$ から, $x \in X \cap Y\ or\ x \in X \cap Z$

故に $XBOOLE\_0 : def2$ から

$X \cap (Y \cup Z) \subseteq X \cap Y \cup X \cap Z$

(1) の証明終了

x を任意にとり, $x \in X \cap Y \cup X \cap Z$ を仮定すると

$XBOOLE\_0 : def2$ から $x \in X \cap Y\ or\ x \in X \cap Z$

$XBOOLE\_0 : def3$ から $x \in X\ and\ x \in Y\ or\ x \in X\ and\ x \in Z$

$XBOOLE\_0 : def2$ から　$x \in X\ and\ x \in Y \cup Z$

故に $XBOOLE\_0 : def3$ と (1) から

$$X \cap (Y \cup Z) = X \cap Y \cup X \cap Z$$

証明終了

1.3.26 theorem BOOLE'71: :: BOOLE'71:

```
theorem BOOLE'71: :: BOOLE'71:
  X \/ Y /\ Z = (X \/ Y) /\ (X \/ Z)
proof
  thus X \/ Y /\ Z c= (X \/ Y) /\ (X \/ Z)
  proof let x;
    assume x in X \/ Y /\ Z;
    then x in X or x in Y /\ Z by XBOOLE_0:def 2;
    then x in X or x in Y & x in Z by XBOOLE_0:def 3;
    then x in X \/ Y & x in X \/ Z by XBOOLE_0:def 2;
    hence thesis by XBOOLE_0:def 3;
  end;
```

```
  let x;
  assume x in (X \/ Y) /\ (X \/ Z);
   then x in X \/ Y & x in X \/ Z by XBOOLE_0:def 3;
   then (x in X or x in Y) & (x in X or x in Z) by XBOOLE_0:def 2;
   then x in X or x in Y /\ Z by XBOOLE_0:def 3;
  hence thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$X \cup Y \cap Z = (X \cup Y) \cap (X \cup Z)$$

証明

先ず次が成り立つ

(1)　$X \cup Y \cap Z \subseteq (X \cup Y) \cap (X \cup Z)$

(1) の証明

x を任意にとり, $x \in X \cup Y \cap Z$ を仮定すると

$XBOOLE\_0 : def2$ から $x \in X$ or $x \in Y \cap Z$

$XBOOLE\_0 : def3$ から $x \in X$ or $x \in Y$ and $x \in Z$

$XBOOLE\_0 : def2$ から $x \in X \cup Y$ and $x \in X \cup Z$

故に $XBOOLE\_0 : def3$ から,

$X \cup Y \cap Z \subseteq (X \cup Y) \cap (X \cup Z)$

(1) の証明終了

次に x を任意にとり, $x \in (X \cup Y) \cap (X \cup Z)$ を仮定すると

$XBOOLE\_0 : def3$ により　$x \in X \cup Y$ and $x \in X \cup Z$

$XBOOLE\_0 : def2$ により $(x \in X$ or $x \in Y)$ and $(x \in X$ or $x \in Z)$

$XBOOLE\_0 : def3$ により $x \in X$ or $x \in Y \cap Z$

故に $XBOOLE\_0 : def2$ により

$(X \cup Y) \cap (X \cup Z) \subseteq X \cup Y \cap Z$

証明終了

1.3.27 theorem :: BOOLE'72:

```
theorem :: BOOLE'72:
```

```
  (X /\ Y) \/ (Y /\ Z) \/ (Z /\ X) = (X \/ Y) /\ (Y \/ Z) /\ (Z \/ X)
proof
  thus X /\ Y \/ Y /\ Z \/ Z /\ X
        = (X /\ Y \/ Y /\ Z \/ Z) /\ (X /\ Y \/ Y /\ Z \/ X) by BOOLE'71
       .= (X /\ Y \/ (Y /\ Z \/ Z)) /\ (X /\ Y \/ Y /\ Z \/ X) by BOOLE'64
       .= (X /\ Y \/ Z) /\ (X /\ Y \/ Y /\ Z \/ X) by BOOLE'69
       .= (X /\ Y \/ Z) /\ (X /\ Y \/ X \/ Y /\ Z) by BOOLE'64
       .= (X /\ Y \/ Z) /\ (X \/ Y /\ Z) by BOOLE'69
       .= (X \/ Z) /\ (Y \/ Z) /\ (X \/ Y /\ Z) by BOOLE'71
       .= (X \/ Z) /\ (Y \/ Z) /\ ((X \/ Y) /\ (X \/ Z)) by BOOLE'71
       .= (X \/ Y) /\ ((Y \/ Z) /\ (X \/ Z) /\ (X \/ Z)) by BOOLE'67
       .= (X \/ Y) /\ ((Y \/ Z) /\ ((X \/ Z) /\ (X \/ Z))) by BOOLE'67
       .= (X \/ Y) /\ (Y \/ Z) /\ (Z \/ X) by BOOLE'67;
end;
```

これは以下の通りです。

$$(X \cap Y) \cup (Y \cap Z) \cup (Z \cap X) = (X \cup Y) \cap (Y \cup Z) \cap (Z \cup X)$$

証明

$$\begin{aligned}
X \cap Y \cup Y \cap Z \cup Z \cap X &= (X \cap Y \cup Y \cap Z \cup Z) \cap (X \cap Y \cup Y \cap Z \cup X) \quad (BOOLE'71 \text{ による}) \\
&= (X \cap Y \cup (Y \cap Z \cup Z)) \cap (X \cap Y \cup Y \cap Z \cup X) \quad (BOOLE'64 \text{ による}) \\
&= (X \cap Y \cup Z) \cap (X \cap Y \cup Y \cap Z \cup X) \quad (BOOLE'69 \text{ による}) \\
&= (X \cap Y \cup Z) \cap (X \cap Y \cup X \cup Y \cap Z) \quad (BOOLE'64 \text{ による}) \\
&= (X \cap Y \cup Z) \cap (X \cup Y \cap Z) \quad (BOOLE'69 \text{ による}) \\
&= (X \cup Z) \cap (Y \cup Z) \cap (X \cup Y \cap Z) \quad (BOOLE'71 \text{ による}) \\
&= (X \cup Z) \cap (Y \cup Z) \cap ((X \cup Y) \cap (X \cup Z)) \quad (BOOLE'71 \text{ による}) \\
&= (X \cup Y) \cap ((Y \cup Z) \cap (X \cup Z) \cap (X \cup Z)) \quad (BOOLE'67 \text{ による}) \\
&= (X \cup Y) \cap ((Y \cup Z) \cap ((X \cup Z) \cap (X \cup Z))) \quad (BOOLE'67 \text{ による}) \\
&= (X \cup Y) \cap (Y \cup Z) \cap (Z \cup X) \quad (BOOLE'67)
\end{aligned}$$

証明終了

1.3.28 theorem BOOLE'40: :: BOOLE'40:

```
theorem BOOLE'40: :: BOOLE'40:
```

```
  X c= Y implies X /\ Z c= Y /\ Z
proof
  assume A1: X c= Y;
  let x;
  assume x in X /\ Z;
  then x in X & x in Z by XBOOLE_0:def 3;
  then x in Y & x in Z by A1,TARSKI:def 3;
  hence thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$X \subseteq Y \Rightarrow X \cap Z \subseteq Y \cap Z$$

証明

$A1 : X \subseteq Y$ を仮定し, x を任意にとり, さらに
$x \in X \cap Z$ を仮定すると
$XBOOLE\_0 : def3$ から $x \in X \ and \ x \in Z$
$\therefore A1, TARSKI : def3$ から $x \in Y \ and \ x \in Z$
故に $XBOOLE\_0 : def3$ から
$X \cap Z \subseteq Y \cap Z$

よって

$$X \subseteq Y \Rightarrow X \cap Z \subseteq Y \cap Z$$

証明終了

1.3.29　theorem :: BOOLE'41:

```
theorem :: BOOLE'41:
  X c= Y & Z c= V implies X /\ Z c= Y /\ V
proof
  assume A1: X c= Y & Z c= V;
  let x;
  assume x in X /\ Z;
  then x in X & x in Z by XBOOLE_0:def 3;
  then x in Y & x in V by A1,TARSKI:def 3;
```

```
  hence thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$X \subseteq Y\ and\ Z \subseteq V \Rightarrow X \cap Z \subseteq Y \cap V$$

証明

次 $A1 : X \subseteq Y\ and\ Z \subseteq V$ を仮定し,
x を任意にとり, さらに $x \in X \cap Z$ を仮定すると
$XBOOLE\_0 : def3$ から $x \in X\ and\ x \in Z$
$\therefore A1, TARSKI : def3$ から $x \in Y\ and\ x \in V$
故に $XBOOLE\_0 : def3$ から
$X \cap Z \subseteq Y \cap V$

よって

$$X \subseteq Y\ and\ Z \subseteq V \Rightarrow X \cap Z \subseteq Y \cap V$$

証明終了

1.3.30 theorem BOOLE'42: :: BOOLE'42:

```
theorem BOOLE'42: :: BOOLE'42:
  X c= Y implies X /\ Y = X
proof
  assume A1: X c= Y;
  thus X /\ Y c= X by BOOLE'37;
  let x;
  assume x in X;
  then x in X & x in Y by A1,TARSKI:def 3;
  hence thesis by XBOOLE_0:def 3;
end;
```

これは以下の通りです。

$$X \subseteq Y \Rightarrow X \cap Y = X$$

証明

$$\begin{aligned}
&A1 : X \subseteq Y \text{ を仮定すると,}\\
&BOOLE'37 \text{ により } X \cap Y \subseteq X\\
&x \text{ を任意にとり}, x \in X \text{ を仮定すると}\\
&A1, TARSKI : def \text{ から } x \in X \ and \ x \in Y\\
&\text{故に } XBOOLE_0 : def3 \text{ から } X \subseteq X \cap Y
\end{aligned}$$

よって

$$X \subseteq Y \Rightarrow X \cap Y = X$$

証明終了

1.3.31 theorem :: BOOLE'38:

```
theorem :: BOOLE'38:
  X /\ Y c= X \/ Z
proof
  X /\ Y c= X & X c= X \/ Z by BOOLE'31,BOOLE'37;
  hence thesis by BOOLE'29;
end;
```

これは以下の通りです。

$$X \cap Y \subseteq X \cup Z$$

証明

$$\begin{aligned}
&BOOLE'31, BOOLE'37 \text{ から } X \cap Y \subseteq X \ and \ X \subseteq X \cup Z\\
&\text{故に } BOOLE'29 \text{ により}\\
&X \cap Y \subseteq X \cup Z
\end{aligned}$$

証明終了

1.3.32 theorem :: BOOLE'44:

```
theorem :: BOOLE'44:
```

```
  X c= Z implies X \/ Y /\ Z = (X \/ Y) /\ Z
proof assume
A1: X c= Z;
  thus X \/ Y /\ Z c= (X \/ Y) /\ Z
  proof let x;
    assume x in X \/ Y /\ Z;
    then x in X or x in Y /\ Z by XBOOLE_0:def 2;
    then x in X or x in Y & x in Z by XBOOLE_0:def 3;
    then x in (X \/ Y) & x in Z by A1,XBOOLE_0:def 2,TARSKI:def 3;
    hence thesis by XBOOLE_0:def 3;
  end;
  let x;
  assume x in (X \/ Y) /\ Z;
  then x in X \/ Y & x in Z by XBOOLE_0:def 3;
  then (x in X or x in Y) & x in Z by XBOOLE_0:def 2;
  then x in X & x in Z or x in Y /\ Z by XBOOLE_0:def 3;
  hence thesis by XBOOLE_0:def 2;
end;
```

これは以下の通りです。

$$X \subseteq Z \Rightarrow X \cup Y \cap Z = (X \cup Y) \cap Z$$

証明

$A1 : X \subseteq Z$ を仮定すると

(1) $X \cup Y \cap Z \subseteq (X \cup Y) \cap Z$

(1) の証明

x を任意にとり, $x \in X \cup Y \cap Z$ を仮定すると

$XBOOLE\_0 : def2$ により $x \in X$ or $x \in Y \cap Z$

$\therefore XBOOLE\_0 : def3$ により $x \in X$ or $x \in Y$ and $x \in Z$

$\therefore A1, XBOOLE\_0 : def2, TARSKI : def3$ により

$x \in (X \cup Y)$ and $x \in Z$

故に $XBOOLE\_0 : def3$ から

$X \cup Y \cap Z \subseteq (X \cup Y) \cap Z$

(1) の証明終了

x を任意にとり, $x \in (X \cup Y) \cap Z$ を仮定すると

$XBOOLE\_0 : def3$ から $x \in X \cup Y$ and $x \in Z$
$\therefore XBOOLE\_0 : def2$ から $(x \in X$ or $x \in Y)$ and $x \in Z$
$\therefore XBOOLE\_0 : def3$ から $x \in X$ and $x \in Z$ or $x \in Y \cap Z$
故に $XBOOLE\_0 : def2$ から
$(X \cup Y) \cap Z = X \cup Y \cap Z$

よって

$$X \subseteq Z \Rightarrow X \cup Y \cap Z = (X \cup Y) \cap Z$$

証明終了

1.3.33 theorem :: BOOLE'53:

```
theorem :: BOOLE'53:
  (X /\ Y) \/ (X /\ Z) c= Y \/ Z
proof
  now let x;
    assume x in (X /\ Y) \/ (X /\ Z);
    then x in (X /\ Y) or x in (X /\ Z) by XBOOLE_0:def 2;
    then (x in X & x in Y) or (x in X & x in Z) by XBOOLE_0:def 3;
    hence x in Y \/ Z by XBOOLE_0:def 2;
  end;
  hence thesis by TARSKI:def 3;
end;
```

これは以下の通りです。

$$(X \cap Y) \cup (X \cap Z) \subseteq Y \cup Z$$

証明

x を任意にとり, $x \in (X \cap Y) \cup (X \cap Z)$ を仮定すると
$XBOOLE\_0 : def2$ から $x \in (X \cap Y)$ or $x \in (X \cap Z)$
$\therefore XBOOLE\_0 : def3$ から
$(x \in X$ and $x \in Y)$ or $(x \in X$ and $x \in Z)$
従って $XBOOLE\_0 : def2$ から $x \in Y \cup Z$
故に $TARSKI : def3$ から
$(X \cap Y) \cup (X \cap Z) \subseteq Y \cup Z$

証明終了

1.3.34 theorem :: BOOLE'45:

```
BOOLE'45: X \ Y = {} iff X c= Y
proof
  thus X \ Y = {} implies X c= Y
  proof
    assume A1:X \ Y = {};
    let x;
    assume x in X & not x in Y;
    then x in X \ Y by XBOOLE_0:def 4;
    hence contradiction by A1,XBOOLE_0:def 1;
  end;
  assume A2:X c= Y;
  now let x;
    x in X & not x in Y iff contradiction by A2,TARSKI:def 3;
    hence x in X \ Y iff x in {} by XBOOLE_0:def 4,def 1;
  end;
  hence thesis by TARSKI:2;
end;
```

これは以下の通りです。

$$X \setminus Y = \{\} \Leftrightarrow X \subseteq Y$$

証明

先ず (1) $X \setminus Y = \{\} \Rightarrow X \subseteq Y$

(1) の証明

$A1 : X \setminus Y = \{\}$ を仮定し,

x を任意にとり, $x \in X$ and not $x \in Y$ を仮定すると

$XBOOLE\_0 : def4$ から $x \in X \setminus Y$

$A1, XBOOLE\_0 : def1$ から矛盾が生じ

(1) の証明終了

次に, $A2 : X \subseteq Y$ を仮定し,

x を任意にとると

$A2, TARSKI : def3$ から

$x \in X$ and not $x \in Y$ からは矛盾が生じ

従っ て $XBOOLE\_0 : def4, def1$ から

$x \in X \setminus Y \Leftrightarrow x \in \{\}$
故に $TARSKI:2$ から
$X \setminus Y = \{\}$

よって

$$X \setminus Y = \{\} \Leftrightarrow X \subseteq Y$$

証明終了

1.3.35 theorem :: BOOLE'90:

```
theorem :: BOOLE'90:
  X \ Y = Y \ X implies X = Y
proof
   assume A1: X \ Y = Y \ X;
   now let x;
    (x in X & not x in Y) iff x in Y \ X by XBOOLE_0:def 4,A1;
    hence x in X iff x in Y by XBOOLE_0:def 4;
   end;
 hence thesis by TARSKI:2;
end;
```

これは以下の通りです。

$$X \setminus Y = Y \setminus X \Rightarrow X = Y$$

証明

$A1 : X \setminus Y = Y \setminus X$ を仮定すると ：
x を任意にとると, $XBOOLE\_0 : def4, A1$ から
$(x \in X\ and\ not\ x \in Y) \Leftrightarrow x \in Y \setminus X$
$XBOOLE\_0 : def4 x \in X \Leftrightarrow x \in Y$
従っ て $XBOOLE\_0 : def4$ から, $x \in X \Leftrightarrow x \in Y$
故に $TARSKI : 2$ により
$X \Rightarrow X = Y$

よって,

$$X \setminus Y = Y \setminus X \Rightarrow X = Y$$

証明終了

1.3.36 theorem BOOLE'46: :: BOOLE'46:

```
theorem BOOLE'46: :: BOOLE'46:
  X c= Y implies X \ Z c= Y \ Z
proof
  assume A1:X c= Y;
  let x;
  assume x in X \ Z;
  then x in X & not x in Z by XBOOLE_0:def 4;
  then x in Y & not x in Z by A1,TARSKI:def 3;
  hence thesis by XBOOLE_0:def 4;
end;
```

これは以下の通りです。

$$X \subseteq Y \Rightarrow X \setminus Z \subseteq Y \setminus Z$$

証明

$$\begin{aligned}
&A1 : X \subseteq Y \text{ を仮定し}\\
&x \text{ を任意にとり, さらに}\\
&x \in X \setminus Z \text{ を仮定すると,}\\
&XBOOLE_0 : def4 \text{ から } x \in X\ and\ not\ x \in Z\\
&\therefore A1, TARSKI : def3 \text{ から } x \in Y\ and\ not\ x \in Z\\
&\text{故に } XBOOLE_0 : def4 \text{ から}\\
&X \setminus Z \subseteq Y \setminus Z
\end{aligned}$$

よって

$$X \subseteq Y \Rightarrow X \setminus Z \subseteq Y \setminus Z$$

証明終了

1.3.37 theorem BOOLE'47: :: BOOLE'47:

$$X \subseteq Y \Rightarrow Z \setminus Y \subseteq Z \setminus X$$

```
theorem BOOLE'47: :: BOOLE'47:
  X c= Y implies Z \ Y c= Z \ X
proof
```

```
  assume A1:X c= Y;
  let x;
  assume x in Z \ Y;
  then x in Z & not x in Y by XBOOLE_0:def 4;
  then x in Z & not x in X by A1,TARSKI:def 3;
  hence thesis by XBOOLE_0:def 4;
end;
```

これは以下の通りです。
証明

$A1 : X \subseteq Y$ を仮定し,
x を任意にとり, さらに
$x \in Z \setminus Y$ を仮定すると,
$XBOOLE\_0 : def4$ により $x \in Z$ and not $x \in Y$
$\therefore A1, TARSKI : def3$ により $x \in Z$ and not $x \in X$
故に $XBOOLE\_0 : def4$ により
$Z \setminus Y \subseteq Z \setminus X$

よって

$$X \subseteq Y \Rightarrow Z \setminus Y \subseteq Z \setminus X$$

証明終了

1.3.38 theorem BOOLE'86:

```
BOOLE'86: X \ (Y /\ Z) = (X \ Y) \/ (X \ Z)
proof
  thus X \ (Y /\ Z) c= (X \ Y) \/ (X \ Z)
  proof
    let x;
    assume x in X \ (Y /\ Z);
    then x in X & not x in (Y /\ Z) by XBOOLE_0:def 4;
    then x in X & (not x in Y or not x in Z) by XBOOLE_0:def 3;
    then x in (X \ Y) or x in (X \ Z) by XBOOLE_0:def 4;
    hence thesis by XBOOLE_0:def 2;
  end;
  Y /\ Z c= Y & Y /\ Z c= Z by BOOLE'37;
  then (X \ Y) c= X \ (Y /\ Z) & X \ Z c= X \ (Y /\ Z) by BOOLE'47;
```

```
  hence (X \ Y) \/ (X \ Z) c= X \ (Y /\ Z) by BOOLE'32;
end;
```

これは以下の通りです。

$$X \setminus (Y \cap Z) = (X \setminus Y) \cup (X \setminus Z)$$

証明

$X \setminus (Y \cap Z) \subseteq (X \setminus Y) \cup (X \setminus Z)$
証明
x を任意にとり,
$x \in X \setminus (Y \cap Z)$ を仮定すると,
$XBOOLE\_0 : def4$ により $x \in X$ and not $x \in (Y \cap Z)$
$\therefore XBOOLE\_0 : def3$ により $x \in X$ and $(not$ $x \in Y$ or not $x \in Z)$
$\therefore XBOOLE\_0 : def4$ により $x \in (X \setminus Y)$ or $x \in (X \setminus Z)$
故に $XBOOLE\_0 : def2$ から $X \setminus (Y \cap Z) \subseteq (X \setminus Y) \cup (X \setminus Z)$
逆に $BOOLE'37$ から, $Y \cap Z \subseteq Y$ and $Y \cap Z \subseteq Z$
$\therefore BOOLE'47$ から $(X \setminus Y) \subseteq X \setminus (Y \cap Z)$ and $X \setminus Z \subseteq X \setminus (Y \cap Z)$
従っ て $BOOLE'32$ から $(X \setminus Y) \cup (X \setminus Z) \subseteq X \setminus (Y \cap Z)$

よって,

$$X \setminus (Y \cap Z) = (X \setminus Y) \cup (X \setminus Z)$$

証明終了